
HP-33S Calculator Program Mathematics 1

Roots of a Quadratic Equation

Programmer: Dr. Bill Hazelton

Date: November, 2007.

Line Instruction User Instructions
R001 LBL  R
R002 SF  10 (Enter using  FLAGS  1  .0)
R003 QUADRATIC ROOTS  EQN  RCL Q  RCL U, etc., ENTER to end.
R004 PSE
R005 ENTER  A  EQN  RCL E  RCL N, etc., ENTER to end.
R006 PSE
R007 INPUT  A
R008 ENTER  B  EQN  RCL E  RCL N, etc., ENTER to end.
R009 PSE
R010 INPUT  B
R011 ENTER  C  EQN  RCL E  RCL N, etc., ENTER to end.
R012 PSE
R013 INPUT  C
R014 RCL  B
R015 x2

R016 RCL  A
R017 RCL  C
R018 ×
R019 4
R020 ×
R021 –
R022 STO  D
R023 x < 0?
R024 VIEW  D If no real roots, program stops here (see Note 2)
R025 √x
R026 RCL–  B
R027 RCL÷  A
R028 2
R029 ÷
R030 STO  P
R031 1ST  ROOT  EQN  1  RCL S  RCL T SPACE, etc.
R032 PSE
R033 VIEW  P
R034 RCL  D
R035 x = 0?
R036 RTN If one real root, program stops here (see Note 3)
R037 √x
R038 RCL+  B
R039 RCL÷  A

Line Instruction User Instructions
R040 2
R041 ÷
R042 +/–
R043 STO  Q
R044 2ND  ROOT  EQN  2  RCL N  RCL D SPACE, etc.
R045 PSE
R046 VIEW  Q
R047 CF  10 (Enter using  FLAGS  2  .0)
R048 RTN

Notes

(1) A simple program to calculate the real roots of a quadratic equation in standard
form:  ax2 + bx + c = 0.

(2) If there are no real roots, the program stops and shows a negative value in the D
storage location. The user should them use the value in D to calculate the
complex roots of the equation, if needed.

(3) If there is a single real root for the equation, the value in storage location D is
equal to zero, so after showing 1ST ROOT, then the value in storage location P,
the program ends.

(4) If there are two real roots, the program continues, displays 2ND ROOT, then
shows the value in storage location Q.

(5) This program sets Flag 10, so that equations can be displayed as prompts. If the
program finishes before displaying two real roots, Flag 10 remains set. If the
program completes, then Flag 10 is cleared.

(6) The program was written to minimize branching, and so the number of labels.

Theory

The solution to a quadratic equation in the form ax2 + bx + c = 0 is:

� 

x = −b ± b2 − 4ac
2a

where  

� 

b2 − 4ac  is known as the discriminant. The discriminant, commonly termed ∆, is stored
in storage location D.

If ∆ is positive, the equation has two real roots. If ∆ = 0, the equation has a single, real root, i.e., it
is a perfect square. If ∆ is negative, the equation has two complex roots. Solution of the complex
roots requires the user to calculate these by hand. The program supplies the value of ∆ in D.

Running the Program

Press  XEQ  R.

Program displays  QUADRATIC ROOT  briefly, then  ENTER  A  briefly, then prompts:
A?
Key in the value of a from the equation. Press R/S.

Program displays  ENTER  B  briefly, then prompts:
B?
Key in the value of b from the equation. Press R/S.

Program displays  ENTER  C  briefly, then prompts:
C?
Key in the value of b from the equation. Press R/S.

If there are no real roots, the program displays:
D =
and the value of the discriminant, ∆, which will be negative. This indicates that there are no real
roots. Press the C key (also the ON/OFF key) to leave the program. (If you press R/S, you will get
a SQRT(NEG) error message. Press the C key to clear the message and end the program.)

If there are real roots, the program displays  1ST  ROOT  briefly, then shows:
P =
and the value of the first root of the equation.
Press R/S to continue.

If there was just one real root, the program now ends and returns to the normal display.
The one root will be in the upper line of the display, and the lower line will show zero.

If there was a second real root, the calculator displays  2ND  ROOT  briefly, then shows:
Q =
and the value of the second root of the equation.
Press R/S to end the program.
The two roots will be shown in the display.

Sample Computations
1. Solve 4x2  +  x  +  8  =  0

There are no real roots, so the program displays  D =   –127.0000

2. Solve 4x2  +  8x  –  6  =  0

There are two real root for this equation.

Solutions: 1st Root P = 0.581139, 2nd Root Q = –2.581139

3. Solve x2  +  2x  + 1  =  0

There is just one root for this equation.

Solution: 1st Root P = –1.00000



HP-33S Calculator Program Mathematics   1
Roots of a Quadratic Equation

- 4 -

Storage Registers Used

A Value of a in equation

B Value of b in equation

C Value of c in equation

D Discriminant, ∆

P Solution of the first root

Q Solution of the second root

Statistical Registers: not used

Labels Used

Label R Length  =   232 Checksum  =  DDFB

Use the length (LN=) and Checksum (CK=) values to check if program was entered correctly.
Use the sample computation to check proper operation after entry.


